

Creating a DOS Batch File to Run SAS® Programs

David Franklin, TheProgrammersCabin.com, Litchfield, NH

ABSTRACT

We often have many SAS programs to run in a directory. While it is possible to run each individually, it is better if a DOS Batch file be created with the list of programs being run and the order in which they are run.

This paper looks at a SAS macro that will take the list of SAS programs in a directory, as specified by the user, and create a DOS Batch file to which can then be run to run all the SAS programs. Also presented will be a small SAS program that you can run at the end of the program to send you an email saying when the programs had finished, and whether there are any issues in the SAS LOG to review!

INTRODUCTION

There are 750 SAS programs in the directory. You could sit there all day and select Batch Submit from Windows Explorer, but that would be a waste of your time. Would it not be better to create a DOS Batch file and set that running? Also, at the end would you not like to know when the batch job finished and report any issues that need your attention?

CREATING THE DOS BATCH FILE

Often I have a number of SAS programs to run so I create a batch file in Windows and then run that batch file. But how do you create that batch file? As nearly always, there is more than one way to do it.

Let's first look at a SAS program which reads in a directory listing of SAS programs (those with the extension .SAS) and then creates a batch file with all the programs.

The first step of the process is to switch on the option NOXWAIT and set the directory we are interested in running the programs, as shown below:

```
%let dir=%str(directory_being_processed);
options NOXWAIT;
```

An example of such a directory_being_processed value could be

```
C:\TUAI\LISTINGS
```

Next step is to create a list of .SAS files and put that in the directory_being_processed directory with the file name of TMP.TXT -- these are in name order, using the option '/o:n' and exclude the directory name in front of the file using the option '/b'.

```
data _null_;
  length _txt $200;
  _txt='dir /b /o:n "||strip("&dir")||'\*.sas" > "||strip("&dir")||'\tmp.txt";
  put _txt=;
  call system(_txt);
run;
```

The following step reads in the names of the SAS programs.

```
data _lst0;
  length fn $200;
  infile "&dir.\tmp.txt";
  input fn;
  fn=upcase(FN);
run;
```

In the above step it is possible to have file name selection clauses after the last statement to exclude any programs that you do not want added to the batch file.

The last step creates the batch file, with the name of the file being specified in the FILE statement:

```
data _null_;
  file "&dir.\RUNALL.bat";
  length logfn txt $200;
  set _lst0 end=eof;
  if _n_=1 then do;
```

```

put "set SASLOC=C:\Program Files\SAS\SAS 9.1";
txt='set PRJLOC='||strip("&dir");
put txt;
txt=scan("&dir",1,"\");
put txt;
put 'CD %PRJLOC%';
put 'DEL /Q *.LOG';
put 'DEL /Q *.LST';
end;
txt="%SASLOC%\sas.exe" -rsasuser -noicon -nosplash '||
'-nostatuswin -nologo -SYSIN "%PRJLOC%\'||strip(fn)||"';
put txt;
if eof then put 'EXIT';
run;

```

It is important to note that the line

```
put "set SASLOC=C:\Program Files\SAS\SAS 9.1";
```

sets the location of the SAS.EXE file which is different for site, operating system and SAS version -- the SET statement is a DOS command that allows a DOS variable to contain a value. A similar task is done in the next line

```
txt='set PRJLOC='||strip("&dir");
```

where the location of the directory being processed is being set.

The following lines will set the drive where your programs are and location for the programs to run:

```

txt=scan("&dir",1,"\");
put txt;
put 'CD %PRJLOC%';

```

Lastly, any LOG and LST files in the directory are deleted using the DOS DEL command:

```

put 'DEL /Q *.LOG';
put 'DEL /Q *.LST';

```

Note that as a rule I delete any LOG files before I run a SAS program just in case the (rare) situation arises where the SAS LOG is not replaced during the run of the program.

The line

```

txt="%SASLOC%\sas.exe" -rsasuser -noicon -nosplash '||
'-nostatuswin -nologo -SYSIN "%PRJLOC%\'||strip(fn)||"';
put txt;

```

creates the DOS SAS batch file statement that calls each program.

Finally, the

```
EXIT
```

command closes the DOS window when the batch file is run.

EMAILING IF THE PROGRAMS RAN OKAY OR HAD ISSUES?

To determine if a SAS program ran okay, it is important to review the SAS LOGs for the run. This is best done via a SAS log checker, an example of which is given below:

```

*Program Name: SASQCLOG.SAS;
* Bring in SAS LOG(s);
data _allogs0;
  attrib _txt length=$256 informat=$char256. format=$char256.
         _fn length=$256
         _myinfile length=$256
         _ln length=8
         _myinline length=8;
  infile "mydirectory\*.log"
         lrecl=256 filename=_myinfile line=_myinline length=len;

```

```

input _txt $varying256. len;
_fn=scan(strip(_myinfile),-1,'\');
_ln=_n_;
proc sort data=_alllogs0;
  by _fn _ln;
data _alllogs0;
  retain _x 0;
  set _alllogs0;
  by _fn _ln;
  drop _x;
  if first._fn then _x=1;
  else _x=_x+1;
  _ln=_x;
run;

* Flag anything questionable;
data _alllogs1;
  retain anyissue 0; *If anything found;
  set _alllogs0 end=eof;
  if index(_txt,'ERROR:') or
    index(_txt,'WARNING:') then _flg=1;
  else _flg=0;
  if index(_txt,'Preload will have no effect.') then _flg=0;
  *Set to no issue, even if exists;
  if _flg=1 then anyissue=1;
  if eof then call symput('anyissue',compress(put(anyissue,8.))); *Something in the LOGS;
run;

* Count the number of issues within each log;
proc summary data=_alllogs1 nway;
  class _fn;
  var _flg;
  output out=_alllogs2 sum=_flgsum;
run;

* Setup for listing;
data _alllogs3;
  set _alllogs2 (in=a where=( _flgsum=0))
    _alllogs1 (in=b where=( _flg=1));
  if a then _txt='** NO ISSUES FOUND **';
proc sort data=_alllogs3;
  by _fn _ln;
options ls=134 ps=55;
title1 "QC of SAS LOG(s)";
run;
proc print data=_alllogs3 noobs label;
  by _fn;
  var _txt;
  label _txt='ISSUES'
        _fn='Program Name';
run;

```

The code above will look for the words 'ERROR:' and 'WARNING:' – if they are found then the lines that contain these words is output to the listing, else if no issues are found then the message "*** NO ISSUES FOUND ***" is displayed.

To add the piece that will email if any issues were found, and if so what they were, it is important to know if any issues were found – this is the ANYISSUE macro value above, 0 if none, 1 if issues. Now incorporate this into an email and we would use the following code:

```

*Send report in email;
filename outmsg email;
data _null_;
  file outmsg
    to="TheBigBoss@company.com"
    cc=("DepartmentAManager@company.com" "DepartmentBManager@company.com")
    bcc="MySelf@company.com"
    subject="Run of Listing for &sysdate9.@&stime."
    attach = "C:\TUAI\LISTINGS\SASQCLOG.LST";
  put 'Dear All';
  put;
  put 'Attached is SASQCLOG report for the TUAI Listings.';
  put;
  if "&anyissue"="1" then
    put "Issues were found in the SAS LOG and needs review.";
  else put "No issues were found in the SAS LOG.";
  put;
  put 'MySelf and I,';
  put 'Section Boss';
run;

```

Finally, the SAS macro that creates the DOS Batch file should be amended so that the program SASQCLOG.SAS is run the end.

CONCLUSION

When the SAS macro that creates the DOS Batch file, it will have the SAS programs that are run in order, as well as the SASQCLOG program which includes the code to email the result. To actually run the DOS batch file, this can be done in several ways, but my far the fastest method is just click, or in some cases double click, the RUNALL.bat file.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: David Franklin
 Enterprise: TheProgrammersCabin.com
 E-mail: dfranklin@TheProgrammersCabin.com
 Web: <http://www.TheProgrammersCabin.com>
 Twitter: ThePgmsCabin

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.